
ABSTRACT
Have you ever asked, “Why doesn't my PDF output look just like my HTML output?” This paper explains the power
and differences of each destination. You will learn how each destination works and understand why the output looks
the way it does. Learn tips and tricks for how to modify your SAS® code to make each destination look more like the
other. The tips span from beginner to advanced in all areas of reporting. Each destination is like a superhero, helping
you transform your reports to meet all your needs. Learn how to use each ODS destination to the fullest extent of its
powers.

INTRODUCTION
Superheroes are all different. Some superheroes have science and technology on their side (Ironman, Batman,
Aquaman); some have super powers because they are from another planet or time (Thor, Superman); while others
became accidental superheroes (Hulk, Spiderman). Just as superheroes are able to do marvelous things, in different
ways, based on their abilities, each ODS destination has a set of abilities and superpowers based on the underlying
architecture and purpose of the destination. This paper addresses two ODS destinations: PDF and HTML.

CREATION STORY
Every superhero has a creation story. PDF and HTML have creation stories too. HTML was probably the first ODS
destination. With SAS version 7, ODS HTML and ODS RTF were initially introduced along with the ODS PRINTER
destination. The PDF destination was not production until SAS 8.2. The underlying architecture of ODS dictated that
style information (colors, fonts, borders, and so on) appropriate to each destination would be sent to each destination
(using an ODS style template) along with the data from the SAS process or procedure. Then ODS would write
instructions that could be rendered by a 3rd party application.
ODS HTML initially created HTML 3.2 compliant HTML tags or elements, in SAS 7. The rendering application for
HTML output was a web browser (although Microsoft Word and Microsoft Excel have been able to open HTML files
ever since Office 97). Initially, with ODS the Printer family consisted of ODS PRINT, ODS PS and ODS PCL. When
you wanted a PDF file, initially, you created a PostScript file and needed to distill it to PDF form. In SAS 8.2, the ODS
PDF destination became production.
Right from the beginning the underlying assumptions of HTML and PDF were different. HTML was designed for
screen viewing. PDF could be viewed on the screen but it creates output that is in a format that is readable by Adobe
or other 3rd party products that consume PDF. Basically it creates output that is printable. What you see in a viewer
like Acrobat will print out with the same appearance. Another nice thing about the PDF destination was that it was a
good way to deliver output that would be hard to change without using fancy Adobe editing tools. HTML was difficult
to change, too, but mostly because not everyone knew HTML tags and instructions. But, HTML is not a “paged”
destination, so things like page numbers and page breaks and printing control really work better in PDF than in HTML
(where they are not used at all). The rendering browser controls how an HTML file will be printed, and to some extent
you might be able to impact that printing using CSS @media instructions.

WHAT IS MEASURED OUTPUT?
The underlying architecture of PDF versus HTML has far-reaching impact on the output, beyond the concept of which
application renders the output. Now we can look at a concrete example. In the code below, two separate reports are
created, taking all the defaults and changing orientation. Other changes are minor. HTML uses the default
HTMLBLUE style template and PDF uses the PRINTER style. By default PDF also includes a bookmark area, but
that feature is easily turned off with the NOTOC option. All the reports use SASHELP.CARS, which is delivered with
Base SAS, so it should be easy to replicate these results.

options orientation=portrait topmargin=.25in bottommargin=.25in
leftmargin=.25in rightmargin=.25in number;
ods html file='c:\temp\default1.html' style=htmlblue;

ods pdf file='c:\temp\default1_portrait.pdf' style=printer notoc;
proc report data=sashelp.cars nowd;
title 'Default Output Portrait Orientation';
footnote 'The Footnote';
run;
ods _all_ close;
options orientation=landscape topmargin=.25in bottommargin=.25in
leftmargin=.25in rightmargin=.25in number;
ods html file='c:\temp\default2.html' style=htmlblue;
ods pdf file='c:\temp\default2_portrait.pdf' style=printer notoc;
proc report data=sashelp.cars nowd;
title 'Default Output Landscape Orientation';
footnote 'The Footnote';
run;
ods _all_ close;
The HTML output is the same for both programs, as shown in Display 1. The reason that both outputs are the same
is that HTML does not use the ORIENTATION option. Everything for HTML is written to a single HTML file, which
represents one (1) web “page”. It is somewhat odd that the HTML specification has uses the term “page” to describe
what is being displayed, since a single web page could be 5000 observations long, definitely too much to be printed
on one physical page. And, as the annotations in red point out, on the HTML page, there is a single title at the top of
the browser display and a single footnote at the bottom of the browser display.

Display 1. Partial HTML Results
However, with PDF output, the concept of measured output comes into play. What is measured output in regard to
ODS PDF? “Measured output” means ODS will determine the amount of available space for output. Unlike HTML
the PDF destination is very concerned about height and width of output objects such as tables, text, graphs, and
images. A good analogy is to think about a piece of paper. How much output can you fit on that paper? The height
and width of the paper is fixed. Therefore, we have to make sure our output fits perfectly. ODS has to take
measurable factors into account, including, but not limited to, paper-size being used, system margins, orientation of
the output and font size of the output. This means that Printer family output, including ODS PDF is bound by physical
limitations.

For example, the Portrait output is 26 pages and the Landscape output is 18 pages. What’s the reason for the
difference? In the Portrait output, the results do not all fit on a hypothetical piece of 8 ½ x 11 paper. As shown in
Display 2, the PDF output displays variables Make -- Invoice on page 1 and then for that same group of observations,
displays variables EngineSize -- Length on page 2. On the other hand, with the landscape orientation, all of the
variables can be displayed in landscape mode. Therefore, the page count is different.
The other concept of measured output is shown in both Display 2 and Display 3. This output shows how ODS
measured the placement of the report rows, titles, and footnotes based on the fact that every character and report
row and text string in the output must be accounted for when ODS PDF lays out the page in memory before writing to
the output file (either the default file or the file you specify with the FILE= option).

Display 2. Partial PDF Results Orientation=Portrait
When you download the programs that accompany this paper, you will see that we also applied a LABEL to some of
the variables to make the Header cells smaller, such as using ‘HP’ for HorsePower and ‘DT’ for DriveTrain. Notice
how the SAS title and footnote statements were used in the measured PDF output. The title and footnote appears at
the top and bottom of each page, as they would, if you routed this output to a physical printer. In addition, the
NUMBER system option was used to place page numbers in the PDF (but not the HTML) output.

Display 3. Partial PDF Results Orientation=Landscape
Now that we have shown some fundamental differences between HTML and PDF output, we can discuss more about
how to take control of your output, beyond simple tricks like changing the labels.

HOW TO SIZE YOUR OUTPUT
Since PDF adheres to the concept of measured output, there is more you can do with PDF in terms of controlling the
size of your output to maximize space on a physical page. For HTML, it is not so critical to maximize space on a
physical page, since the concept of physical pages is not relevant to HTML. When we show some of the controls that
apply to PDF, we will also show how those changes impact the HTML output.
One easy change to impact PDF output is to change margins and orientation as shown in the Display 2 and 3 and
this will, in turn, impact number of pages that “fit” in each orientation. Other ways exist to impact the size of the
output, one such is font size. We can make two simple changes to the PROC REPORT output to show how font size
and cell padding can impact output. Consider the following change to the PROC REPORT code:

proc report data=sashelp.cars nowd
style(report)={fontsize=9pt cellpadding=2px}
style(header)={fontsize=9pt}
style(column)={fontsize=9pt};
As shown in Display 4, the PDF output, now fits in 8 portrait pages instead of the 26 pages before. The reason for this
difference is that those 2 simple changes allowed ODS to “measure” the PDF output differently, so that all the
variables in the report row would fit into one portrait page. Although landscape output is not shown, using these same
style overrides with PDF output caused the number of landscape pages to shrink to 11 pages instead of the original
18 pages.

Display 4. Partial PDF Results with Font and Cell Padding Changes
The idea of changing font size and making it smaller might be an obvious change, but why does cell padding work?
Cell padding is the amount of white space that “cushions” the letters in the cell. To see the impact of cell padding, we
can change the cell padding to be a very big number (like 20 px) and then look at the output again, as shown in
Display 5.

Display 5. Partial PDF Results with Font and Large Cell Padding Changes
With this unusually large value for cell padding, the report row is again, too wide to fit on one portrait page and the
number of total pages in the output has increased to 40. These considerations do have an impact on HTML, as
shown in Display 6, but not as dramatic as increasing or decreasing the number of pages, because the only thing that
happens is that you scroll a bit more or a bit less in the browser. Even though there are no page breaks, you can use
these techniques with HTML to impact how much content fits on a single screen.

Display 6. Partial HTML Results with Font and Cell Padding Changes

PAGES OF OUTPUT (OR WHAT IF MY OUTPUT IS TOO TALL)
So far, the control over page breaking has been “implicit” or implied paging. When there are too many report rows to
fit on a page (in paged destinations), a new page of output is started. The SAS titles and footnotes will take up space
on every page in paged destinations, like PDF, but will appear only at the top and bottom of the output table in HTML.
We can move outside the world of the simple listing report and talk about explicit page breaking. An explicit page
break is one that is inserted in procedure output by procedure controls, such as using the PAGE dimension in PROC
TABULATE or the PAGE option in PROC REPORT or BY and PAGEBY with PROC PRINT. Since ODS destinations,
except for LISTING, do not use LINESIZE and PAGESIZE options.
One simple way to break up the output is to add explicit page breaks using procedure controls. To that end, we will
start with a switch of procedures and show implicit page breaks with PROC TABULATE and then explicit page breaks
with a few other procedures.
The code that we are starting with is shown below:

ods html file='c:\temp\demo2_implicit_page.html' style=htmlblue;
ods pdf file='c:\temp\demo2_implicit_page.pdf' style=printer;
proc tabulate data=sashelp.cars;
title '1) Implicit Page Break from Procedure';
where make in ('Audi', 'Volvo', 'BMW', 'Chevrolet') and
type in ('Sedan', 'Wagon');
class make model type;
var msrp mpg_highway mpg_city;
table make * model,
type*mean*(msrp mpg_highway mpg_city);
run;
ods _all_ close;
There are so many values for MAKE and MODEL that the TABULATE output is difficult to read, as shown in Display
7.

Display 7. PDF Results from PROC TABULATE with Implicit Page Breaks
But, with a change to the table statement to create a PAGE dimension results in the output shown in Display 8.

table make,
model,
type*mean*(msrp mpg_highway mpg_city);

Display 8. PDF Results from PROC TABULATE with Explicit Page Breaks
Using the same page dimension technique with HTML causes a slightly different result. Using the default HTMLBLUE
style, there is a horizontal rule at the logical page break. In addition, the SAS title repeats at the top of the table and, if
there were a footnote, the footnote would appear under the table on each logical page.

Display 9. HTML Results from PROC TABULATE with Explicit Page Breaks
If you have a browser that supports CSS (Cascading Style sheets), you might be grateful that the horizontal rule is
there. In the style template, the horizontal rule comes with a CSS instruction for page breaking:
class html
"Common HTML text used in the default style" /
'expandAll' = "<span onclick=""expandCollapse()"">"
'posthtml flyover line' = "</span><hr size=""3""/>"
'prehtml flyover line' = "<span><hr size=""3""/>"
'prehtml flyover bullet' = %nrstr("<span><b>&#183;</b>")
'posthtml flyover' = "</span>"
'prehtml flyover' = "<span>"
'break' = "<br/>"
'Line' = "<hr size=""3""/>"
'PageBreakLine' =
"<p style=""page-break-after: always;""><br/></p><hr size=""3""/>"
'fake bullet' = %nrstr("<b>&#183;</b>");

This means that when you print from a browser, such as Internet Explorer, if the CSS command is respected, the
page break command will be used by the browser, as shown in Display 10.

Display 10. HTML Results Displayed in Internet Explorer’s Print Preview Mode
However, once the style is changed to a different style like SEASIDE, the horizontal rule disappears as shown in
Display 11, but the title and footnote repeat for every logical page.
Of course, the quickest way to generate either physical or logical pages with either PDF or HTML is to use BY-group
processing. Consider Display 12, a listing report that uses a BY statement added to PROC REPORT. Just remember
that anytime you use BY-group processing the data must be sorted or indexed on the BY variables.

proc sort data=sashelp.cars out=cars;
by make type;
where make in ('Audi', 'Volvo', 'BMW', 'Chevrolet') and
type in ('Sedan', 'Wagon');
run;
ods html file='c:\temp\demo2_use_by.html' style=htmlblue;
ods pdf file='c:\temp\demo2_use_by.pdf' notoc style=printer;
proc report data=cars nowd;
by make type;
run;
ods _all_ close;

Display 11. HTML Results from PROC TABULATE Using SEASIDE Style

Display 12. PDF Results from PROC REPORT Using BY-Group Processing

Even if you add BY-group processing, however, remember that other factors, such as font size, margin options and
orientation can still impact your output.
Another way to impact the information that appears on a page is to control the number of report columns that appear.
There are several ways to do this. One way that works only for PDF (or RTF) is to use the COLUMNS= option of the
PDF destination. COLUMNS= is a useful option when you have enough room on the physical page to display the
report rows in multiple column format. The COLUMNS= option produces the output shown in Display 13. You use the
COLUMNS= option on the ODS destination statement, as shown below.

ods pdf file='c:\temp\demo3_use_columns.pdf' notoc style=printer columns=2;
proc report data=cars nowd;
column make model msrp;
by type;
. . . more code . . .
run;
ods _all_ close;
Display 13 shows how the BY group for TYPE=WAGON started a new page in the PDF results.

Display 13. PDF Results Using the COLUMNS= Option
The HTML destination does not support the COLUMNS= option. However, there is an ODS destination, called ODS
TAGSETS.HTMLPANEL that will allow you to produce “paneled” output as shown in Display 14.
TAGSETS.HTMLPANEL allows you to specify the number of “panel rows” and “panel columns” using “event”
instructions

ods tagsets.htmlpanel style=htmlblue
file='c:\temp\demo3_use_HTMLPANEL.html'
options(panelcolumns='2' doc='help' panelborder='2'
embedded_titles='No');
ODS tagsets.htmlpanel event=row_panel(start);
. . . proc report code . . .
ods _all_ close;

Display 14. HTML Results Using the HTMLPANEL Destination
These last two examples allow you page control when your report information is narrow enough to fit multiple
columns in a page. But, read on to find out what you do if your output truly is too wide.

WHAT IF MY OUTPUT IS TOO WIDE
Having output too wide is not a concern for HTML output. As mentioned above PDF has a fixed width and height, so
having data too wide is a possibility. Here is a simple PROC PRINT that shows output that is too wide.

ods pdf file='panel.pdf' style=printer;
proc print data=sashelp.cars;
run;
ods _all_ close;

12

Display 15. PDF Results Showing Paneled Output
As you can see, this data set contains 16 columns. ODS PDF determines that all columns will not fit in the available
space. By using internal calculations it determines how many columns are within the page of output. Again I want to
reiterate that when we say “page of output” that does not mean a physical page. A page of output is what ODS PDF
calculates as the height and width available to show the output. When ODS PDF recognizes that 16 columns will not
fit on the page, it determines how many will. A secondary calculation occurs that determines how many rows can fit
on one page while still showing all the columns. The result is what we called “paneling” in ODS PDF. Looking at the
above example you can see ODS PDF determined 16 observations showing all 16 columns could fit on one page. In
order to fit all 16 columns, we will display a split table on the page. We display the first 9 columns of 16 observations
on the top half of the page. Then we display the remaining 7 columns. You will also notice that PROC PRINT
repeated the “OBS” column to assist you in matching up the split table by observation. Not only did our SAS code
produce output too wide, it is also too tall, so we will have some pagination.

Display 16. Continued PDF Results Showing Paneled Output

Looking at output you can see how the bottom of page 1 completes the first 16 observations of the table. ODS PDF
will start the next 16 observations on page 2. This will continue until all observations of the table are complete. In the
above example we got 27 pages of output. Paneled output normally is not the preferred way of viewing your data. It
can get confusing and cumbersome to for your data to be fragmented and displayed over many pages.
If you do want to control paneling, you have to look to procedures like PROC REPORT to help you in this regard. You
can control which variable starts the new panel and which variables repeat across panels through a use of the ID
option and the PAGE option. For example, if you use the code below:

proc report data= cars nowd spanrows
style(report)={width=100%};
title '1) Explicit Paneling';
column Make Type Model Origin DriveTrain MSRP Invoice EngineSize Cylinders
Horsepower MPG_City MPG_Highway Weight Wheelbase Length;
define make / order id;
define type / order id;
define Cylinders / page;
break after make/page ;
run;
In this code, the values of MAKE and TYPE will repeat for every panel and the first panel will be composed of all the
variables starting with MAKE and ending with ENGINESIZE; the second panel will contain MAKE and TYPE
(repeated from the first panel), followed by the variables CYLINDERS through LENGTH, as listed in the COLUMN
statement. This technique works for PDF, as shown in Display ~17x~.

Display 17. Explicit Paneling with PROC REPORT in PDF Output
What you notice in the PDF output is how the WIDTH=100% style override made both panels stretch from margin to
margin. Then, the ID option in the DEFINE statement ensured that MAKE and TYPE were repeated for every panel.
Finally, the PAGE option in the DEFINE statement for CYLINDERS caused the second panel to start with the
CYLINDERS variable after the repeat of MAKE and TYPE. Notice how each panel appears on a separate page. We
can control this using the STARTPAGE= option, by adding STARTPAGE=NO to the ODS PDF invocation statement.
(In our program, we used a WHERE statement to limit the MAKE value to Mazda, in order to generate only one page
of output, as shown in Display ~18x~.

ods pdf file='c:\temp\demo4_explicit_panel_startpage.pdf' style=printer
startpage=no;
proc report data= cars nowd spanrows style(report)={width=100%};
where make = 'Mazda';
. . . rest of code same as previous code . . .
ods _all_ close;

The HTML output that is created using PROC REPORT and explicit paneling only can be compared to Display ~17x~
because HTML is not a “paged” destination and there is no STARTPAGE option available for the HTML destination.
However, you can see in Display ~19x~, that the panels in the HTML output are divided onto logical “pages” which
we can see by the use of the horizontal rule between each panel.

Display 18. Explicit Paneling with PROC REPORT in PDF Output Using STARTPAGE=NO

Display 19. Explicit Paneling with PROC REPORT in HTML Output
While explicit paneling can also help if you have too many variables to fit on a printed page in the PDF destination,
the fact still remains that paneled output is not always the best way to display your data. It might be that some of the
changes we have already shown will be suitable for your purpose better.
If only ODS PDF had scroll bars like ODS HTML has to view the output. The PDF viewer (Acrobat Reader) does have
a scroll bar, but ODS PDF still is creating an image of the report, as it will look if printed. So, if you do not want explicit
paneling, how do we prevent paneling in ODS PDF? Above we talked about font size, cell padding, orientation, and
margins to control our output. These are all great ideas but sometimes you might have to alter your output to not
display all the available columns. Since paneling and controlling “pages” does not really impact HTML the way it
impacts PDF, it’s time to move on to other ways to control how your output looks.

HOW TO CONTROL THE OUTPUT
In this section, we are going to talk about controlling the output from two different standpoints: controlling the output
with destination-specific options (this will mostly apply to ODS PDF) and controlling the output with style-specific
options (this will apply to both ODS HTML and ODS PDF).
Earlier we discussed explicit and implicit page breaks. There is another way to create explicit breaks that you can
insert yourself but only with the ODS PDF destination. The attribute is called STARTPAGE and as we have seen in
the simple example above, it can easily turn off all page breaks. But, there are other values for STARTPAGE other
than STARTPAGE=NO. In the ODS PDF statement you can enter STARTPAGE=YES|NO|NOW.
Using the option YES (which is the default) after every procedure block an explicit page break will be issued. Using
the option NO means that no page break will be issued and ODS PDF will try to fit as much as possible until an
implicit page break occurs. The last option is NOW, which issues an immediate page break at your request. If you
started your output with STARTPAGE=NO in order to get the two panels for Mazda on one physical page, then you
might use STARTPAGE=NOW to issue page breaks where you need them, such as before you generate the two
panels for Volvo. This type of code can be easily “macro-ized” so that the panels look like multiple tables produced for
all the variables within a group.
The code that produced Display ~20x~ shows using STARTPAGE=NOW to produce 2 pages of paneled output.

ods pdf file='demo4_panel_more_startpage.pdf' style=printer startpage=no;
** PROC REPORT for Mazda;
ods pdf startpage=now;
** PROC REPORT for Volvo;
ods _all_ close;

Display 20. Explicit Paneling with PROC REPORT in HTML Output
But a more in depth discussion of PROC REPORT for paneling falls outside the scope of this paper. Suffice it to say
that beyond implicit and explicit paneling, there is even more in the world of ODS to allow you to control placement of
ODS objects in a specific layout using ODS LAYOUT. But, ODS LAYOUT is a very specialized layout tool and does
not work for ODS HTML at all. Now we can move on to a discussion of what can be controlled using style templates
and style overrides.
Of course, the best way to control output look and feel is with ODS style templates. For example, if both ODS PDF
and ODS HTML use the JOURNAL style, we see the results as shown in Display ~21x~. The JOURNAL style is a
great black and white style that is designed for journal publication. It has a crisp “no interior lines” table look with
black borders between the header columns and the data cell area.
Generally speaking, color, and font changes will end up making HTML and PDF output look the same, as this style
override in PROC REPORT shows.

proc report data=sashelp.cars nowd
style(Header)={font_face='Albany AMT' font_size=12pt
color=cx004c1a background=cxccccff}
style(column)={font_face='Albany AMT' font_size=11pt}
style(summary)={font_face='Albany AMT' font_size=14pt
color=cx004c1a background=cxccccff};
column make model ('Average' msrp invoice mpg_city mpg_highway);
define make / group
style(column)=Header{font_face='Albany AMT' font_size=12pt
color=cx004c1a background=cxccccff};
define model / group

define
define
define
define
rbreak
run;

style(column)=Header{font_face='Albany AMT' font_size=12pt
color=cx004c1a background=cxccccff};
msrp / mean;
invoice / mean;
mpg_city/mean f=comma6.;
mpg_highway / mean f=comma6.;
after / summarize;

These style override methods in PROC REPORT have the same effect on both outputs, as shown in Display ~22x~.

Display 21. Using the Same Style Template for PDF and HTML Output

Display 22. Using STYLE= Statement-level Overrides for PDF and HTML Output

One challenge that people always worry about is how to make cells a specific height or width and another challenge
is how to flow long text within a cell. As it turns out, all of these challenges can be met by using STYLE overrides. The
CELLHEIGHT attribute is good when you want to call attention to one report row in particular.
The code to use CELLHEIGHT is illustrated in this example with PROC REPORT, changes to the previous program
were minimal, so only the changed statements are shown below.

proc report data=sashelp.cars nowd
. . . more code . . .
style(summary)={font_face='Albany AMT' font_size=14pt
color=cx004c1a background=cxccccff
cellheight=1.25in vjust=m};
. . . more code . . .
compute msrp;
if msrp.mean gt 80000 then do;
call define(_col_,'style','style={cellheight=.75in vjust=m}');
end;

Display 23. Changing CELLHEIGHT for PDF and HTML Output
The output shown in Display ~23x~ shows that both destinations treated the SUMMARY line as expected, with an
increased CELLHEIGHT and a vertical alignment of the middle of the cell. However, what we see for the result of the
CALL DEFINE conditional style change is that both destinations will not make just one cell in a row higher than the
cells in the rest of the row. So the CELLHEIGHT change impacted the entire row. However, PDF treated the
VJUST=M override as an instruction to impact the whole row. While for the HTML destination, the VJUST=M had an
impact on only the MSRP cell, as coded. So, if all we do is change the first argument in the CALL DEFINE from
_COL_ to _ROW_, then we see that both destinations treat all the cells in the report row the same, as shown in
Display ~24x~.

Display 24. Changing Entire Row Height and Vertical Alignment
Sometimes people assume that when they have very long text that the solution is the CELLHEIGHT style attribute,
but that is not the case. With ODS, the height of a cell will increase to automatically fit what is inside the cell. You can
actually impact the height of a cell that is going to contain long text by adjusting the WIDTH or CELLWIDTH attribute.
One thing about both PDF and HTML is that the destinations are smart enough to flow or wrap text within a specified
cell width. For the next example, SASHELP.CARS has been modified to have some phrases by Lewis Carroll added
to some of the values for the MODEL variable. You can see the full code when you download the ZIP file of
programs.
The program that alters the cell width attribute. Note that you need to use CELLWIDTH= for version 9.1.3 or earlier
and you can use WIDTH= or CELLWIDTH= in version 9.2 or later.

proc report data=cars nowd;
title '1a) Narrow Width Causes Wrapping That Influences Cell Height';
column make model type msrp invoice;
define model / style(column)={width=1.0in};
run;
proc report data=cars nowd;
title '1b) Making Model Column Wider Changes Wrapping and Height';
column make model type msrp invoice;
define model / style(column)={width=2.0in};
run;
Notice how, in Display ~25x~, the cell heights for all the data cells have automatically adjusted to accommodate the
defined width of 1”. But in Display ~26x~, which was the result of only changing the CELLWIDTH attribute to 2”, some
of the higher cells in the first example now have enough width to “spread out”. Therefore, the cells are not as high in
report 1a as they were in report 1b.
You might have noticed that some of our code uses WIDTH versus CELLWIDTH or HEIGHT versus CELLHEIGHT or
OUTPUTWIDTH versus WIDTH. The fact is that any version of SAS after 9.2 will use either attribute name. This is
true of other attributes, such as FOREGROUND versus COLOR and BACKGROUND versus
BACKGROUNDCOLOR. But if you are still using version 9.1.3 or earlier, then you need to use the original style
attribute, as documented for your version of SAS. The relevant documentation topic is entitled, “Style Attributes and
Their Values.”

Display 25. Changing Cell Width to “Flow” Long Text

Display 26. Adjusting Width Impacts Cell Height

One of the last style changes we will show is a new (with SAS 9.2) style attribute that can be used for text decoration
to underline, overline, or strike through text. For example, when people use the JOURNAL style, they frequently want
to underline the column header values. In the past, this was something that you could do only with RTF or HTML
using destination-specific syntax. However, starting with SAS 9.2, using the TEXTDECORATION style attribute, this
feature became available for PDF, as well as the other destinations. Consider the following code that produced
Display ~27x~. Not only does it underline the column headers, but it also uses the LINK= option of the SAS TITLE
statement to insert a hyperlink into the REPORT and ODS ESCAPECHAR with the PREIMAGE attribute to insert a
logo into the report.

ods escapechar='^';
proc report data=sashelp.cars(obs=20) nowd
style(report)={width=50%}
style(header)={fontweight=bold fontsize=14pt textdecoration=underline};
where substr(make,1,1) in ('A','B','C');
title j=l '^{style[preimage="c:\sgf2014\create_superhero.png"] }'
j=r 'My Report';
title2 link="http://marvel.com/games/play/31/create_your_own_superhero"
'Create Your Own SuperHero}';
column make model msrp mpg_city;
run;

Display 27. Using ODS ESCAPECHAR and Other Style Changes
What you will notice is that the hyperlink for the Marvel.com site is highlighted differently in PDF (with a blue box)
versus HTML, but both links, when clicked, will take you to a website where you can create your own superhero. For
this paper, Cynthia was HTMLWoman and Scott was PDF Guy.
The final word on controlling your output is that it will all come down to style in the end. Whether you use style
templates, style overrides or ODS ESCAPECHAR, it will be worth your time to invest in a study of ODS styles. Last,
but certainly not least, we are going to focus on our two destinations and how they work with non-tabular output.

NON TABULAR OUTPUT
There are other types of output other than tables in your output. We have mentioned Titles and Footnotes. But there
are other types of non-tabular output like images, SAS/GRAPH, and ODS GRAPHICS we have to think about when
designing our output. Each destination handles these differently. We will discuss images first. Three very common

ways to use images are as a background, pre-image, and post-image. In the ODS PDF destination the DPI (dot per
inch) of the image determines the size that will be displayed. It’s good to remember the higher the image DPI the
more precise the appearance of the image. Here is an example using the same image with different image DPIs.
ods escapechar = '^';
title '^{style [preimage="c:\sgf2014\SAS_logo150.jpg"] my title w/SAS @ 150 DPI}';
ods pdf file="prt_image.pdf";
proc print data=sashelp.class(obs=1);run;
ods text = '^{style [preimage="c:\sgf2014\SAS_logo100.jpg"] text statement w/SAS @ 100
DPI}';
ods _all_ close;

Display 28. Images Displayed within ODS PDF Destination
Notice the pre-image on the title statement has image DPI of 150. The image on the text statement below the table
has image DPI of 100. Understanding the different sizes of the images will help you plan the space you use on the
page. Using the properties of the image file will tell you initial size of the image.

Display 29. Image Property File

The properties of the image will show you the height, width, and DPI of the image. Using simple math you can divide
the height and width by the DPI to calculate its size.
Width will be 500 / 150 = 3.33 inches
Height will be 170 / 150 = 1.19 inches
The other image had an image DPI of 100. That image has a larger height and width since its divisor is only 100
rather than 150.
ODS PDF has a default DPI equal to 150. When ODS PDF encounters an image that is not 150 DPI, it will perform
some internal calculations to scale the image. The second image will grow bigger since the image DPI was less than
the PDF’s default DPI. A good guideline is to always use images with a DPI of 150 to ensure no surprises occur
when placing an image inside your PDF file. Of course using images with multiple image DPIs within one output will
result in scaling.
If you cannot change your image DPI, you can always set the default DPI for ODS PDF by using the “DPI=n” (where
n is the DPI of your choice) attribute on your ODS PDF statement.
ods pdf file="prt_image.pdf" dpi=100;
Now ODS HTML handles images much differently that ODS PDF. Here is the output of ODS HTML using the same
program.

Display 30. Images Displayed within ODS HTML Destination

The image is exactly the same no matter what the image DPI is. HTML does not worry about image DPI. It knows
the height and width from the properties of the image. Both images are exactly 500 pixels x 179 pixels. Monitors
have PPI (pixels per inch), so the ‘per inch’ does not consider DPI. It is just displaying the pixels value, which is the
same regardless of DPI. If the monitor has a higher PPI, the image might be smaller. The DPI only affects the clarity
of the image, not the size.
There are other non-tabular outputs like SAS/GRAPH and ODS GRAPHICS. Again both destinations handle them in
different ways. ODS HTML handles graphs very similar in the way it handles images. The graph is saved off in an
image format and HTML just references that file when needed. ODS PDF embeds the graph inside the PDF file. The
sizing is dependent on the graph sizing options. Now, we can look at SAS/GRAPH and see how it appears in ODS
PDF. Here is a simple PROC GPLOT.
goptions reset=all border;
title 'notice graph is full size of page';
ods pdf file="graph.pdf" notoc;
proc gplot data=sashelp.class; plot age*height; run;
ods pdf close;

Display 31. SAS/GRAPH Displayed within ODS PDF Destination
I did not show the whole page but the GPLOT does fit the entire page in ODS PDF. Graph objects are embedded
into the PDF. ODS PDF and the GRAPH subsystem communicate about available space before placing down the
output. Graph with no size setting will always want a full page for the output. Now we can do another example with a
table and graph on same page.
options number;
goptions reset=all border;
title 'GPLOT is full size';
ods pdf file="graph.pdf" notoc startpage=no;
proc print data=sashelp.class(obs=5); run;
proc gplot data=sashelp.class; plot age*height; run;
ods pdf close;

Display 32. SAS/GRAPH Displayed within ODS PDF Destination (page 1 of 2)
Page 1 is complete with a small table with 5 observations. Notice since the graph wanted a full page it is positioned it
on the next page even though I said “startpage=no” in the ODS PDF statement.

Display 33. SAS/GRAPH Displayed within ODS PDF Destination (page 2 of 2)
Most people like to maximize space in their output and want the graph to be on the same page as the table.
GOPTIONS is what we need to solve this problem. GOPTIONS are graphical options that can control your graphical
output. We can set two size attributes (hsize and vsize) to make the graph fit in a smaller space. Now we can try it
one more time with GOPTIONS set.
options number;
goptions reset=all border hsize=5in vsize=5in;
title 'GPLOT is smaller';
ods pdf file="graph.pdf" notoc startpage=no;
proc print data=sashelp.class(obs=5); run;
proc gplot data=sashelp.class; plot age*height; run;
ods pdf close;

Display 34. SAS/GRAPH Displayed within ODS PDF Destination All on 1 Page

So using GOPTIONS the graph is telling ODS PDF that it only needs 5 inches of both height and width. ODS PDF
determines that space is available on page 1. Therefore, it places it below the table.
As I said before HTML just references the file that SAS/GRAPH creates and displays it on the page.

goptions reset=all border;
title 'Graph within HTML';
ods html file="graph.html";
proc print data=sashelp.class(obs=5);run;
proc gplot data=sashelp.class; plot age*height; run;
ods html close;

Display 35. SAS/GRAPH Displayed within ODS HTML Destination
Remember I said HTML just references the file and uses the size in the file’s properties. Here is the file that was
created in my current working directory by SAS/GRAPH.

Display 36. Property File of GPLOT (initial size)
If I want to use GOPTIONs to set my size of the gplot, the PNG will have different dimensions and again HTML just
plugs the file into the output. Here are the properties if I set hsize and vsize on the GPTIONS.
goptions reset=all border hsize=5in vsize=5in;

Display 37. Property File of GPLOT (size set by GOPTIONS statement)
Notice how the width and height changed. SAS/GRAPH uses 96 DPI when creating the graphs. Using simple math
again we see the 480/96 = 5. We set the vsize and hsize to 5 inches in our SAS program. Now we can see how
SAS/GRAPH honored our settings and created the graph to our specifications.
Now we can look at ODS GRAPHICS. It will be the same as SAS/GRAPH. ODS HTML will look for the created file
in the current work directory containing the graph. ODS PDF will check space availability before placing the graph in
the output. There are two differences with ODS GRAPHICS:
(1) It has a different statement to set its height and width. It uses the ODS GRAPHICS statement.
(2) The default size in PDF will NOT be the whole page. ODS GRAPHICS has a default size set in a template.
Here is an ODS GRAPHICS example in ODS PDF.
ods graphics on / reset=all;
ods pdf file="sgplot.pdf" style=styles.harvest;
title1 'ODS Graphics work the same way';
proc sgplot data=sashelp.class;
vbar age;
run;
ods graphics off;
ods pdf close;

Display 38. ODS GRAPHICS Output within ODS PDF Destination
I used a style called harvest to help show that the SGPLOT is not taking up the whole page on page 1. Using the
ODS GRAPHICS statement you can set a height and width. As mentioned above, this statement allows you to size
your graph and maximize your space in your output.
options number;
ods graphics on / reset=all height=5in width=5in;
ods pdf file="sgplot1.pdf" startpage=no;
title1 'ODS Graphics work the same way';
proc sgplot data=sashelp.class;
vbar age;
run;
proc print data=sashelp.class(obs=5);run;
ods graphics off;
ods pdf close;

Display 39. ODS GRAPHICS Output within ODS PDF Destination (all on one page)
If we printed out this page, and used a ruler to measure the SGPLOT we would see the graph is 5 inches by 5 inches.
It’s nice to know that ODS PDF will give you graphs the exact why you specify it. Knowing this allowing you to size
your graphs based on your tabular output. You can fit more on page if you know how big the output will be.
Non-Tabular output (images, SAS/GRAPH, and ODS GRAPHICS) can present challenges especially in the ODS
PDF destination. The good news is there are ways to resize this output and make it fit as you desire.

CONCLUSION
Life and our comic books would be boring if all the superheroes were the same. Part of the fun of reading about or
watching superheroes is seeing how creative each one of them is when faced with an obstacle. Sure, Superman can
just blow a boulder off of a railroad track, but Batman can shift the boulder with gadgets and Wonder Woman might
have to use her golden lasso and some scientific leverage. Can ODS PDF be similar to ODS HTML and vice versa?
The answer is sometimes.
Just as we are happy about the differences between our superheroes, we need to learn that each destination has
things that it does well (like dealing with page numbers and page breaks for PDF or dealing with wide tables like
HTML) and other things that it does not do as well. But if you remember that ODS HTML is the output used by web
browsers while ODS PDF is measured output used for printing, then you will gradually learn to appreciate each
destination for what it does best. People will always ask how to make one destination look like the other. We think
the real answer is each destination is perfect for what it does best. No one likes to print their output from a web page.
No one thinks a piece of page can hold all their data at once. We have given you several tools within each
destination to modify its appearance. You control the output now. You know what to expect and how to use the tools
to solve the problems. So there you go, leaping tall destinations with a single bound!

To help you out in your superhero destination adventures, we have included some bonus programs in the ZIP file of
programs to illustrate some of the things that each destination does best.
Paper 038-2014

CONCLUSIONS
Table 1 shows the summary of techniques and their usage in each destination:
Technique or Option

PDF

HTML

+
+

-

COLUMNS=

+

(not an option
for HTML)

Implicit Page Breaks

+
+

-

Papersize, Printer options
Margin, Orientation, Date, Number
Options

Explicit Page Breaks

+ (logical page
breaks use
horizontal rule)

Font-related style change
Cellpadding style change
Color-related style changes
Graph sizing options

+
+
+
+

+
+
+
+

Table 1. Comparison of Techniques

